```
-- DiskKD.Mesa  Edited by Sandman on August 23, 1977  9:42 PM

DIRECTORY
  AltoDefs: FROM "altodefs",
  AltoFileDefs: FROM "altofiledefs",
  BootDefs: FROM "bootdefs",
  DirectoryDefs: FROM "directorydefs",
  DiskDefs: FROM "diskdefs",
  DiskKDDefs: FROM "diskkddefs",
  InlineDefs: FROM "inlinedefs",
  ImageDefs: FROM "imagedefs",
  SegmentDefs: FROM "segmentdefs";

DEFINITIONS FROM AltoDefs, AltoFileDefs, SegmentDefs;

DiskKD: PROGRAM
  IMPORTS BootDefs, DirectoryDefs, DiskDefs, ImageDefs, SegmentDefs EXPORTS DiskKDDefs =

  BEGIN

  InitializeDiskKD: PUBLIC PROCEDURE =
    BEGIN
    pages: PageCount;
    IF ~DirectoryDefs.DirectoryLookup[@diskKD.file.fp, nameKD, FALSE]
      THEN SIGNAL FileNameError[nameKD];
    MoveFileSegment[diskKD, DefaultBase, 1];
    OpenDiskKD[];
    DiskDefs.SetDisk[@kd.disk];
    pages ← (kd.size+PageSize-1)/PageSize;
    [] ← CloseDiskKD[];
    MoveFileSegment[diskKD, DefaultBase, pages];
    RETURN
    END;

  OpenDiskKD: PROCEDURE =
    BEGIN
    IF ~diskKD.swappedin THEN
      BEGIN  SwapIn[diskKD];
      kd ← FileSegmentAddress[diskKD];
      kd.changed ← 0;
      END
    ELSE swapKD.proc ← CantSwap;
    RETURN
    END;

  UpdateDiskKD: PUBLIC PROCEDURE =
    BEGIN
    IF diskKD.swappedin
    AND kd.changed#0 THEN
      BEGIN
      diskKD.write ← TRUE;
      SwapUp[diskKD];
      diskKD.write ← FALSE;
      kd.changed ← 0;
      END;
    RETURN
    END;

  CloseDiskKD: PUBLIC PROCEDURE RETURNS [BOOLEAN] =
    BEGIN
    IF ~diskKD.swappedin THEN RETURN[FALSE];
    swapKD.proc ← CantSwap;  UpdateDiskKD[];
    Unlock[diskKD];  SwapOut[diskKD];
    RETURN[TRUE]
    END;

  CleanupDiskKD: PUBLIC ImageDefs.CleanupProcedure =
    BEGIN
    SELECT why FROM
      Finish, Abort, OutLd =>  [] ← CloseDiskKD[];
      -- Save =>
      --   We depend on MakeImage to call CloseDiskKD when
      --   it has finished allocating the image file pages.
      --   Logically, it should also call ResetDisk at this
      --   time, but it can't do that until the Restore.
      -- Restore =>
```

```
      --   We depend on MakeImage to call InitializeDiskKD
      --   as soon as the image file starts up so that the
      --   Real to Virtual disk address map can be set up.
      ENDCASE;
    RETURN
    END;


  AllOnes: WORD = 177777B;

  NewSN: PUBLIC PROCEDURE RETURNS [sn:SN] =
    BEGIN  OpenDiskKD[];
    IF (kd.lastSN.part2 ← kd.lastSN.part2+1) = 0
      THEN kd.lastSN.part1 ← kd.lastSN.part1+1;
    sn ← kd.lastSN;  kd.changed ← AllOnes;
    swapKD.proc ← CloseDiskKD;
    RETURN
    END;

  BitAddress: TYPE = RECORD [word:[0..7777B], bit:[0..17B]];

  DiskFull: PUBLIC SIGNAL = CODE;

  AssignDiskPage: PUBLIC PROCEDURE [da:vDA]
    RETURNS [vDA] =
    BEGIN OPEN InlineDefs;
    onebit: WORD;
    ba, wa: CARDINAL;
    w: POINTER TO WORD;
    base: BitAddress = LOOPHOLE[da+1];
    baseWa: CARDINAL ← base.word;
    baseBa: CARDINAL ← base.bit;
    OpenDiskKD[];
    DO ENABLE UNWIND => swapKD.proc ← CloseDiskKD;
      FOR wa IN [baseWa..kd.size) DO
        IF (w ← @kd.table[wa])↑ # AllOnes THEN
          FOR ba IN [baseBa..wordlength) DO
            onebit ← BITSHIFT[100000B,-ba];
            IF BITAND[w↑,onebit]=0 THEN
              BEGIN
              w↑ ← BITOR[w↑,onebit];
              kd.changed ← AllOnes;
              swapKD.proc ← CloseDiskKD;
              RETURN[vDA[wa*wordlength+ba]];
              END;
            ENDLOOP;
        baseBa ← 0;
        ENDLOOP;
      IF baseWa=0 THEN SIGNAL DiskFull;
      baseWa ← 0;
      ENDLOOP;
    END;

  ReleaseDiskPage: PUBLIC PROCEDURE [v:vDA] =
    BEGIN OPEN InlineDefs;
    word: POINTER TO WORD;
    OpenDiskKD[];
    word ← @kd.table[LOOPHOLE[v,BitAddress].word];
    word↑ ← BITAND[word↑,
      BITNOT[BITSHIFT[100000B,-LOOPHOLE[v,BitAddress].bit]]];
    kd.changed ← AllOnes;
    swapKD.proc ← CloseDiskKD;
    RETURN
    END;

  CountFreeDiskPages: PUBLIC PROCEDURE RETURNS [count: CARDINAL] =
    BEGIN OPEN InlineDefs;
    ba, wa: CARDINAL;
    onebit: WORD;
    word: POINTER TO WORD;
    count ← 0;
    OpenDiskKD[];
    FOR wa IN [0..kd.size) DO
      IF (word ← @kd.table[wa])↑ # AllOnes THEN
        FOR ba IN [0..wordlength) DO
          onebit ← BITSHIFT[100000B,-ba];
```

```
        IF BITAND[word↑,onebit]=0 THEN count ← count+1;
        ENDLOOP;
    ENDLOOP;
  swapKD.proc ← CloseDiskKD;
  RETURN
  END;


-- Main Body

kd: POINTER TO KD;
swapKD: SwapStrategy;
diskKD: FileSegmentHandle;
cleanupKD: ImageDefs.CleanupItem;
nameKD: STRING = "DiskDescriptor.";

swapKD.proc ← CantSwap;
cleanupKD.proc ← CleanupDiskKD;
diskKD ← NewFileSegment[BootDefs.BootFile[Read+Write],DefaultBase,1,Read];
AddSwapStrategy[@swapKD];
ImageDefs.AddCleanupProcedure[@cleanupKD];
InitializeDiskKD[];

-- Should we support running without a DiskDescriptor?

END.
```